J 研究会資料　2006/2/25

Ｊのオブジェクト指向プログラミングーその３

ＪのＧｒｉｄでカレンダーを作って遊ぶ

西川　利男

表計算（スプレッドシート）として、一般にはＥｘｃｅｌがよく使われている。しかし、ＪのＧｒｉｄはそれに劣らず、Ｊユーザにとっては、内部が分かるだけ融通性があり、便利で強力なツールである。

昨年暮れの JAPLA シンポジウムでは、筆者はＪのオブジェクト指向（ＯＯＰとそれによるＧｒｉｄプログラミングの適用例につき紹介した[1]。今回は、カレンダーを作って表示する課題を例として、ＪのＧｒｉｄプログラミングをやってみよう。

１．年月日から曜日を得る計算とカレンダーの作成

西暦の年月日から曜日を求めるには、次のゼラーの公式が知られている[2]。西暦の年月日を $(100C+Y)$ 年 (M) 月 (D) 日とするとき次の式を用いる。ここで[ ]はガウスの記号（整数部を取り出す）である。なお、１月、２月はそれぞれ前の年で $M=13, M=14$ とする。

$$W = \left[\frac{21C}{4}\right] + \left[\frac{5Y}{4}\right] + \left[\frac{26(M+1)}{10}\right] + D - 1 \pmod 7$$

ここで得た W の値 0, 1, 2, …,6 に応じて、日、月、火、…、土と曜日が求められる。

このような計算はＪではべつに何ということもなく、以下のように簡単に行われるが、Excel だけの計算ではちょっと難しいだろう。

これを元にカレンダーを表形式で作る。これもＪなど配列言語ではいとも簡単である。

文献

[1] 西川利男「Ｊのオブジェクト指向プログラミングーその１、その２」
JAPLA シンポジウム, 2005/12/10

[2] 藤村幸三郎、田村三郎「数学歴史パズル」講談社ブルーバックス p.211 (1935)

曜日を求めるプログラム week とその実行例は次のようになる。

```
week =: 3 : 0
C =. <. (0{y.)% 100
Y =. 100 | 0{y.
M =: 1{y.
if. M < 3
  do.
    M =: M + 12
    Y =. Y - 1
  end.
D =. 2{y.
W =. 7 |(<. (21*C)%4) + (<. (5*Y)%4) + (<.(26*(M+1))%10) + (D - 1)
> W{'Sun';'Mon';'Tue';'Wed';'Thu';'Fri';'Sat'
)
   week 2006 4 27
Thu
```

カレンダー作成のプログラム calendar とその実行例は次のようになる。

```
calendar =: 3 : 0
W=:week y. , 1
select. 1{y.
  case. 1; 3; 5; 7; 8; 10; 12 do. MD =: 31
  case. 4; 6; 9; 11 do. MD =: 30
  case. 2 do. if. 0 = 4 | (0{y.) do. MD =: 29 else. MD =. 28 end.
end.
(5, 7)$(W#_), (>: i. MD), 7#_
)

   calendar 2006 2
 _  _  _  1  2  3  4
 5  6  7  8  9 10 11
12 13 14 15 16 17 18
19 20 21 22 23 24 25
26 27 28  _  _  _  _
```

ここでは、J の数値としての無限大(_)をうまく利用した。

２．カレンダー表示のＧｒｉｄプログラミング

Ｇｒｉｄ（スプレッドシート）のプログラムを一から作るとなると大変なことで、一般のユーザにはほとんど不可能である。しかし、ＪではＯＯＰのクラスとして提供されているので、それを利用すればそれほどのことはない。

Ｊのクラス jwatch にはスプレッドシートのほとんどの機能が備わっている。これはまたいろいろなマウス操作の基本のクラス jwgrid, jzgrid を継承しているので jwatch のインスタンスを作るだけでこれらの処理が簡単に行える。

Ｇｒｉｄでカレンダーを表示するには、次のようにするだけでよい。

```
corequire ‘jwatch’
CAL =: calendar 2006 2
w1 =: ‘CAL’ conew ‘jwatch’
```

マウス、キーボードなどを使ってＧｒｉｄの操作をいろいろやってみよう。

・ CTRL-e セルへの値の入力可能の切り替え

セルへの値の入力はそのままではロックされていて入れられない。上のようにすると、入力が可能となる。

・ CTRL-f フォントの変更

いろいろフォントを変えることもできる。

・ CTRL-c コピー

・ CTRL-v 貼り付け

これはマウスでセルを範囲指定して行うこともできる。

・ CTRL-m 式の入力と処理結果の色表示

これを使って以下のような計算式を入力するとおもしろい結果が現れる。

```
1 = 2| y.              値が偶数のセルに色がつく
primch_base_ y.     値が素数のセルに色がつく
```

ここで、素数のチェックを行うために、以下のようなプログラムを用いた。

```
primch =: 3 : 0"(0)
if. y. = _
       do. 0
       else. 1 = (#@:q:) y.
end.
)
```

なお、Ｇｒｉｄの中から元のスクリプトファイルのプログラムを呼ぶときの参照方法に注意のこと。単に primch では参照されない。 primch_base_ として、関数の所在を示さなければならない。

カレンダーとするには、最初の行に'Sun', 'Mon', 'Tue', … のように曜日を表示したいものだが、セルに上のような文字列を入力しようとしても出来ない。理由はこのＧｒｉｄの値は数値データを想定しているからである。したがって、つぎには文字列の値を持つようにＧｒｉｄを作り変えてみよう。

文字列で行うためには、以下のような変換が必要になる。

```
CALC =:  “: L:0  <” 0 _ , CAL
W2 =:  ‘CALC’   conew   ‘jwatch’
```

こうすれば、曜日の文字列の値が入れられる。当然のことながら、日付の数字に対してはそのままでは計算は出来ない。そこで、先の素数チェックのプログラムを以下のようにちょっと修整した primchc を作る。

```
primchc =: 3 : 0
if. # ". y. do. primch ". y. else. 0 end.
)
```

こうして、CTRL-m で、次のように数式を入れれば、前と同様、素数のセルは色付きで示される。

```
   >primchc_base_  each  y.
```

Ｊ－Ｇｒｉｄによるカレンダーの実行例は次のようになる。



最後に、マウスで全体を選択した上で、CTRL-c でコピーを行ってから、Excel を開いてここに貼り付けてみると、あざやかに Excel シート上にカレンダーが出来上がる。このように Windows のクリップボードを介してデータを共有することも可能である。

```
NB. Calendar Display on Grid
NB.   by T. Nishikawa 2006/2/13
NB. Calculate Week Day by Zeller's Formula
wr=: 1!:2&2

NB. usage => week 2006 4 27
week =: 3 : 0
C =. <. (0{y.)% 100
Y =. 100 | 0{y.
M =: 1{y.
if. M < 3
   do.
      M =: M + 12
      Y =. Y - 1
   end.
D =. 2{y.
W0 =. (<. (21*C)%4) + (<. (5*Y)%4) + (<.(26*(M+1))%10) + (D - 1)
W =: 7 | W0
> W{'Sun';'Mon';'Tue';'Wed';'Thu';'Fri';'Sat'
)

NB. usage => calendar 2006 2
calendar =: 3 : 0
week y. , 1
select. 1{y.
   case. 1; 3; 5; 7; 8; 10; 12 do. MD =: 31
   case. 4; 6; 9; 11           do. MD =: 30
   case. 2                     do. if. 0 = 4 | (0{y.) do. MD =: 29 else. MD =. 28 end.
end.
(5, 7)$(W#_), (>: i. MD), 7#_
)

NB. Grid Display Program
corequire 'jwatch'

NB. Calendar by Grid for Number
```

```
caln =: 3 : 0
Calendar_N =: calendar y.
w1 =: 'Calendar_N' conew 'jwatch'
)

NB. Calendar by Grid for Character String
calc =: 3 : 0
CALN =. calendar y.
CALC =. '': L:0 <"0 CALN
Calendar_C =: ('Sun';'Mon';'Tue';'Wed';'Thu';'Fri';'Sat'), CALC
w2 =: 'Calendar_C' conew 'jwatch'
)

NB. Prime Check 0: No, 1: Yes
primch =: 3 : 0"(0)
if. y. = _
   do. 0
   else. 1 = (#@:q:) y.
end.
)
```